

Generating Random Numbers

Random numbers are numbers that we can't predict. Flipping a coin or rolling dice will give us a random number. Random numbers are very important in games and in some kinds of math. Computers can generate random numbers pretty well. QBasic's `RND` function provides random numbers that we can use.

RND

`RND` is a special function that give us a random number between 0 and 1. We can use this in games to make things interesting. `RND` is perfect for rolling dice or flipping a coin. First let's see `RND` in action:

```
CLS
PRINT RND
PRINT RND
```

This program will print `RND` (a pseudo random number) twice. Notice that we get two numbers that appear to be unpredictable and random. But, try running the program again. We'll get the same "random" numbers. This means our games would always be the same each time the user runs them. Fortunately, we have a way to fix this problem.

RANDOMIZE TIMER

Using `RANDOMIZE TIMER` will make sure the random numbers we get are different each time we run `RND`. The only thing which continuously changes in the computer is its internal clock. The `RANDOMIZE TIMER` function returns the number of seconds elapsed since midnight. Try this:

```
CLS
RANDOMIZE TIMER
PRINT RND
PRINT RND
```

Now when we re-run our little program a different set of random numbers will be generated.

Useful Random Numbers

Random numbers between 0 and 1 are nice but they aren't really very useful. What we need for a game might be a random number between 1 and 6, like when we roll dice. To get something more useful, we'll use a little math. Fortunately, computers are very good a math so we won't have to be.

There are two problems we must solve in order to get the results (random numbers between 1 and 6) we want. First, the range of random numbers has to be expanded from 0 through 1 to 1 through 6. That's easily done like this:

```
CLS
RANDOMIZE TIMER
PRINT RND * 6 + 1
PRINT RND * 6 + 1
```

By multiplying by 6, we increase the range to 0 through 5. Then by simply adding 1 to our product we shift the range up one giving us 1 through 6. However, we have another problem. All that decimal stuff on the end. QBasic's built-in INT function is quite useful for converting a decimal number such as this, to an integer (a number without a decimal) like we want for our dice game.

```
CLS
RANDOMIZE TIMER
PRINT INT(RND * 6 + 1)
PRINT INT(RND * 6 + 1)
```

Roll the Dice

Here's a complete program that rolls two dice and print the value of each. The variables `die1` and `die2` are used to hold the values of each die before printing. In a real game, `die1` and `die2` would be used in some clever way to change the outcome of the game.

```
'*****
'
'               RANDOM NUMBER GENERATOR
'               A simple program to roll two dice
'*****

'*****DECLARATIONS*****
DIM die1, die2 AS INTEGER
DIM rollAgain AS STRING

RANDOMIZE TIMER

'*****USER INPUT*****
CLS
INPUT "Press ENTER to roll the dice.", A$

'*****COMPUTATIONS*****
DO
    CLS

    die1 = INT(RND * 6 + 1)
    die2 = INT(RND * 6 + 1)

    PRINT
    PRINT "Die 1: "; die1
    PRINT "Die 2: "; die2
    PRINT

    INPUT "Would you like roll again (Y/N)"; rollAgain
    rollAgain = LEFT$(UCASE$(rollAgain), 1)

LOOP UNTIL rollAgain = "N"

END
```